
Abstract. We present a formal method for component-based system specifica-
tion and verification which is based on the new algebraic specification language
CafeOBJ, which is a modern successor of OBJ incorporating several new devel-
opments in algebraic specification theory and practice.
We first give an overview of the main features of CafeOBJ, including its logi-
cal foundations, and then we focus on the behavioural specification paradigm in
CafeOBJ, surveying the object-oriented CafeOBJ specification and verification
methodology based on behavioural abstraction.
The last part of this paper further focuses on a component-based behavioural
specification and verification methodology which features high reusability of
both specification code and verification proof scores. This methodology consti-
tutes the basis for an industrial strength formal method around CafeOBJ.

1 Overview of CafeOBJ

CafeOBJ (whose definition is given by [7]) is a modern successor of the OBJ language
[18, 10] incorporating several new major developments in algebraic specification the-
ory and practice. It is aimed to be an industrial strength language, suitable both for re-
searchers and for practitioners. This section is devoted to a brief overview of CafeOBJ,
including its main features, its specification and verification environment, and its logical
foundations.

1.1 CafeOBJ Main Features

Equational Specification and Programming. This is inherited from OBJ [18, 10] and
constitutes the basis of the language, the other features being somehow built on top of
it. As with OBJ, CafeOBJ is executable (by term rewriting), which gives an elegant
declarative way of functional programming, often referred as algebraic programming.1

As with OBJ, CafeOBJ also permits equational specification modulo several equational
theories such as associativity, commutativity, identity, idempotence, and combinations
between all these. This feature is reflected at the execution level by term rewriting mod-
ulo such equational theories.

? On leave from the Institute of Mathematics of the Romanian Academy, PO Box 1-764,
Bucharest 70700, ROMANIA.

1 Please notice that although this paradigm may be used as programming, this aspect is still
secondary to its specification side.

Behavioural Specification. Behavioural specification [13, 14, 4] provides another
novel generalization of ordinary algebraic specification but in a different direction. Be-
havioural specification characterizes how objects (and systems) behave, not how they
are implemented. This new form of abstraction can be very powerful in the specification
and verification of software systems since it naturally embeds other useful paradigms
such as concurrency, object-orientation, constraints, nondeterminism, etc. (see [14] for
details). Behavioural abstraction is achieved by using specification with hidden sorts
and a behavioural concept of satisfaction based on the idea of indistinguishability of
states that are observationally the same, which also generalizes process algebra and
transition systems (see [14]).

CafeOBJ directly supports behavioural specification and its proof theory through
special language constructs, such as

– hidden sorts (for states of systems),
– behavioural operations (for direct “actions” and “observations” on states of sys-

tems),
– behavioural coherence declarations for (non-behavioural) operations (which might

be either derived (indirect) “observations” or “constructors” on states of systems),
and

– behavioural axioms (stating behavioural satisfaction).

The advanced coinduction proof method receives support in CafeOBJ via a de-
fault (candidate) coinduction relation (denoted =*=). In CafeOBJ, coinduction can
be used either in the classical HSA sense [14] for proving behavioural equivalence of
states of objects, or for proving behavioural transitions (which appear when applying
behavioural abstraction to RWL).2

Besides language constructs, CafeOBJ supports behavioural specification and ver-
ification by several methodologies.3 CafeOBJ currently highlights a methodology for
concurrent object composition which features high reusability not only of specification
code but also of verifications [7, 8]. Behavioural specification in CafeOBJ might also
be effectively used as an object-oriented (state-oriented) alternative for traditional ADT
specifications. Experiments seem to indicate that an object-oriented style of specifica-
tion even of basic data types (such as sets, lists, etc.) might lead to higher simplicity of
code and drastic simplification of verification process [7].

Behavioural specification is reflected at the execution level by the concept of be-
havioural rewriting [7, 4] which refines ordinary rewriting with a condition ensuring
the correctness of the use of behavioural equations in proving strict equalities.

Rewriting Logic Specification. Rewriting logic specification in CafeOBJ is based
on a simplified version of Meseguer’s rewriting logic [20] specification framework for
concurrent systems which gives a non-trivial extension of traditional algebraic specifi-
cation towards concurrency. RWL incorporates many different models of concurrency

2 However, until the time this paper was written, the latter has not been yet explored sufficiently,
especially practically.

3 This is still an open research topic, the current methodologies might be developed further and
new methodologies might be added in the future.

in a natural, simple, and elegant way, thus giving CafeOBJ a wide range of applica-
tions. Unlike Maude [2], the current CafeOBJ design does not fully support labelled
RWL which permits full reasoning about multiple transitions between states (or system
configurations), but provides proof support for reasoning about the existence of transi-
tions between states (or configurations) of concurrent systems via a built-in predicate
(denoted ==>) with dynamic definition encoding both the proof theory of RWL and the
user defined transitions (rules) into equational logic.

From a methodological perspective, CafeOBJ develops the use of RWL transitions
for specifying and verifying the properties of declarative encoding of algorithms (see
[7]) as well as for specifying and verifying transition systems.

Module System. The principles of the CafeOBJ module system are inherited from
OBJ which builds on ideas first realized in the language Clear [1], most notably institu-
tions [11, 9]. CafeOBJ module system features

– several kinds of imports,
– sharing for multiple imports,
– parameterized programming allowing

• multiple parameters,
• views for parameter instantiation,
• integration of CafeOBJ specifications with executable code in a lower level

language
– module expressions.

However, the theory supporting the CafeOBJ module system represents an updating of
the original Clear/OBJ concepts to the more sophisticated situation of multi-paradigm
systems involving theory morphisms across institution embeddings [5], and the concrete
design of the language revise the OBJ view on importation modes and parameters [7].

Type System and Partiality. CafeOBJ has a type system that allows subtypes based
on order sorted algebra (abbreviated OSA) [17, 12]. This provides a mathematically
rigorous form of runtime type checking and error handling, giving CafeOBJ a syntactic
flexibility comparable to that of untyped languages, while preserving all the advantages
of strong typing.

Since at this moment there are many order sortedness formalisms, many of them
very little different from others, and each of them having its own technical advantages
and disadvantages and being most appropriate for a certain class of applications, we
decided to keep the concrete order sortedness formalism open at least at the level of
the language definition. Instead we formulate some basic simple conditions which any
concrete CafeOBJ order sorted formalism should obey. These conditions come close to
Meseguer’s OSAR [21] which is a revised version of other versions of order sortedness
existing in the literature, most notably Goguen’s OSA [12].

CafeOBJ does not directly do partial operations but rather handles them by us-
ing error sorts and a sort membership predicate in the style of membership equational
logic (abbreviated MEL) [21]. The semantics of specifications with partial operations
is given by MEL.

1.2 The CafeOBJ Specification and Verification Environment

Although this is rather a feature of the current system rather than of the language, due
to its importance for the effective use of the current CafeOBJ system, we briefly survey
it here.

The CafeOBJ system includes an environment supporting specification documents
with formal contents over networks and enabling formal verifications of specifications.
The CafeOBJ environment takes advantage of current InterNet technologies and can
be thought as consisting of four parts:

– The interpreter in isolation acts very much like the OBJ3 interpreter by check-
ing syntax and evaluating (reducing) terms. In addition, the CafeOBJ interpreter
incorporates an abstract TRS machine and a compiler.

– The proof assistant extends the theorem proving capabilities of the interpreter with
more powerful, dedicated provers.

– The document manager takes care of processing of specification documents over
networks.

– Specification libraries focus on several specific problem domains, such as object-
oriented programming, database management, interactive systems, etc.

1.3 CafeOBJ Logical Foundations

CafeOBJ is a declarative language with firm mathematical and logical foundations in
the same way as other OBJ-family languages (OBJ, Eqlog [15, 3], FOOPS [16], Maude
[20]) are. The reference paper for the CafeOBJ mathematical foundations is [6], while
the book [7] gives a somehow less mathematical easy-to-read (including many exam-
ples) presentation of the semantics of CafeOBJ. In this section we give a very brief
overview of the CafeOBJ logical and mathematical foundations, for a full understand-
ing of this aspect of CafeOBJ the reader is referred to [6] and [7].

The mathematical semantics of CafeOBJ is based on state-of-the-art algebraic
specification concepts and results, and is strongly based on category theory and the
theory of institutions [11, 5, 9]. The following are the principles governing the logical
and mathematical foundations of CafeOBJ:

P1. there is an underlying logic4 in which all basic constructs and features of
the language can be rigorously explained.

P2. provide an integrated, cohesive, and unitary approach to the semantics of
specification in-the-small and in-the-large.

P3. develop all ingredients (concepts, results, etc.) at the highest appropriate
level of abstraction.

The CafeOBJ Cube. CafeOBJ is a multi-paradigm language. Each of the main
paradigms implemented in CafeOBJ is rigorously based on some underlying logic;
the paradigms resulting from various combinations are based on the combination of

4 Here “logic” should be understood in the modern relativistic sense of “institution” which pro-
vides a mathematical definition for a logic (see [11]) rather than in the more classical sense.

logics. The structure of these logics is shown by the following CafeOBJ cube , where
the arrows mean embedding between the logics, which correspond to institution embed-
dings (i.e., a strong form of institution morphisms of [11, 9]) (the orientation of arrows
correspond to embedding “less complex” into “more complex” logics).

MSA

HSA

RWL

HSRWL

OSA OSRWL

HOSRWLHOSA H = hidden
A = algebra
O = order
S = sorted
M = many
RWL = rewriting logic

The mathematical structure represented by this cube is that of a lattice of institution
embeddings [5, 6]. By employing other logical-based paradigms the CafeOBJ cube
may be thought as a hyper-cube (see [6, 7] for details). It is important to understand
that th CafeOBJ logical foundations are based on the CafeOBJ cube rather than on its
flattening represented by HOSRWL.5

The design of CafeOBJ lead to several important developments in algebraic spec-
ification theory. One of them is the concept of extra theory morphism [5], which is a
concept of theory morphism across institution embeddings, generalizing the ordinary
(intra) theory morphisms to the multi-paradigm situation. Another important theoreti-
cal development is constituted by the formalism underlying behavioural specification
in CafeOBJ which is a non-trivial extension of classical hidden algebra [14] in several
directions, most notably permitting operations with several hidden arguments via the
crucial coherence property. This extension is called “coherent hidden algebra” in [4]
and comes very close to the “observational logic” of Bidoit and Hennicker [19]. The
details of the “coherent hidden algebra” institution can be found in [6].

2 Behavioural Specification in CafeOBJ

Behavioural specification might be the most distinctive feature of CafeOBJ within the
broad family of algebraic specification languages. As mentioned above, behavioural
specification paradigm is incorporated into the design of the language in a rather direct
way. Also, this paradigm constitutes the core of the current CafeOBJ object-oriented
specification and verification methodologies. We devote this section to a methodologi-
cal presentation of the behavioural specification paradigm in CafeOBJ, trying also to
explain the main concepts behind this paradigm.

5 The technical reason for this is that model reducts across some of the edges (i.e., the left-
to-right ones) of the CafeOBJ cube involve both an ordinary model reduct and a non-trivial
reduct along the corresponding institution embedding, see [6, 5, 7] for details.

2.1 Basic Behavioural Specification

Basic behavioural specification is the simplest level of behavioural specification in
which the operations are either actions or observations on the states of the objects. Let
us consider an object-oriented (or “state-oriented”) CafeOBJ specification for lists:

mod! TRIV+ (X :: TRIV) {
op err : -> ?Elt

}
mod* LIST {
protecting(TRIV+)
*[ List ]*
op nil : -> List
bop cons : Elt List-> List -- action
bop car : List -> ?Elt -- observation
bop cdr : List -> List -- action
vars E E’ : Elt
var L : List
eq car(nil) = err .
eq car(cons(E, L)) = E .
beq cdr(nil) = nil .
beq cdr(cons(E, L)) = L .

}
This is quite different from the usual data-oriented specification of lists. In our be-
havioural specification, lists are treated as objects with states (the sort of states is the
hidden sort List), and the usual list operations (cons and cdr ) act on the states of the list
object or (car ) observe the states. Actions and observations are specified as behavioural
operations. In general, a behavioural operation is called action iff its sort is hidden (i.e.,
state type), and is called observation iff its sort is visible (i.e., data type). Behavioural
operations are restricted to have exactly one hidden sort in their arity, this monadicity
property being characteristic to behavioural operations (either actions or observations).
Behavioural operations define the behavioural equivalence relation between the states
of the object, denoted as ∼:

s ∼ s′ iff [c(s) = c(s′) for all visible behavioural contexts c]

A behavioural context c is any string of behavioural operations (this makes sense be-
cause of the monadicity property on hidden sorts of the behavioural operations). c is
visible iff its sort is visible; this is the same as saying that c has an observation at the
top. It is important to notice that behavioural equivalence is a semantic notion; this
means that whenever we consider a behavioural equivalence relation we need to con-
sider a model (i.e., an implementation) for the specification6.

CafeOBJ methodologies introduce a graphical notation extending the classical ADJ-
diagram notation for data types for behavioural specification in which

G1. Sorts are represented by ellipsoidal disks with visible (data) sorts represented in
white and hidden (state) sorts represented in grey, and with subsort inclusion rep-
resented by disk inclusion, and

6 Which needs not to be a concrete one.

G2. Operations are represented by multi-source arrows with the monadic part from the
hidden sort thickened in case of behavioural operations.

The list specification can be therefore visualised as follows:

List

Elt

?Elt

cdr

car

cons

err

nil

car(nil)  = err
car(cons(E,L)) = E
cdr(nil) ~ nil
cdr(cons(E,L)) ~ L

Several other aspects of this specifications need special attention. The first one con-
cerns the data of this specification and the error handling aspect of this methodology.
LIST specifies a list object over any set of elements. “Any set of elements” is speci-
fied by the built-in module TRIV which specifies one sort (Elt) with loose denotation
(hence its denotation is given by all sets); this is used as a parameter of the specifica-
tion LIST and can be instantiated to any concrete data type. The error handling aspect
arises because of the partiality of car. TRIV+ just introduces a new error element (err ).
The error supersort ?Elt is built-in7 and err is the only new element belonging to [the
denotation of] ?Elt; this is ensured by the free extension of [the loose denotation of]
TRIV which is specified by giving TRIV+ initial denotation (mod!). Notice that this
style of error handling contrasts the more complex data-oriented approach which uses
a subsort for the non-empty lists and overloads the list operations on this subsort. This
methodological simplification is mainly possible because of the loose denotation of be-
havioural specification (with the adequate “loose” behavioural equality) which avoids
the strictness of the initial denotation of the data-oriented approach.

Another aspect is given by the use of behavioural equations in the specification
LIST. Behavioural equations represent behavioural equivalence relations between states
rather than strict equalities. Therefore each model (implementation) of LIST does not
need to interpret cdr(cons(e,l)) as l, where e is an element and l is a list8, but rather as
a state behavioural equivalent to l. For example, if one implements the list object as an
array with pointer, in this model (implementation) this equality does not hold strictly,
but it holds behaviourally. Generally speaking, behavioural equality is the meaningful
equality on hidden sorts, while the strict equality is the meaningful equality for the
visible (data) sorts. However, there are situations when the strict equality on hidden sorts
is also necessary. Behavioural abstraction also provides a nice way of error handling for

7 It is provided by the system.
8 Better said, a state of the list object.

hidden sorts, as shown by the other behavioural equation. Thus instead of introducing a
(hidden) error for cdr(nil), we rather shift the error handling to the data type by saying
this is behaviourally equivalent to nil.9 A finer analysis of the behavioural equivalence
on the list object (see the section below) tells us that the behavioural equality between
cdr(nil) and nil is exactly the same with saying that car (cdr n(nil)) = err for all natural
numbers n, which is the natural minimal condition for the behaviour of nil.

2.2 Behavioural Specification with Hidden Constructors

Behavioural specification with hidden constructors is a more advanced level of be-
havioural specification which relies on the important novel concept of behavioural co-
herence first defined and studied in [7, 4] and which was first realized by the CafeOBJ
language [7].

At the general level, a hidden constructor is an operation on hidden sorts10 whose
sort is also hidden and which is not declared behavioural. This means that such oper-
ation does not take part in the definition of the behavioural equivalence relation. Also
(and related to the above), a hidden constructor need not be monadic on the hidden
sorts, thus it may admit several hidden sorts in the arity.

In the data-oriented specification of lists there is a difference in nature between cons
and cdr, in that cons is a “constructor” and cdr is a “destructor”. This different nature
of cons and cdr reflects in the behavioural specification too and is formally supported
by the fact that one may prove (from the specification LIST) that for all lists l and l′,

l ∼ l′ iff [car(cdr n(l)) = car(cdrn(l′)) for all natural numbers n]

Technically this means that for the purpose of defining the appropriate behavioural
equivalence for lists, cons does not play any rôle, therefore it may be specified as an
ordinary operation, hence cons is a hidden constructor. Consequently, the only real be-
havioural operations are the observation car and the action cdr. This new specification
for lists can be visualized by the following CafeOBJ diagram:

List

Elt

?Elt

cdr

car

cons

err

nil

car(nil)  = err
car(cons(E,L)) = E
cdr(nil) ~ nil
cdr(cons(E,L)) ~ L

9 Recall that in LISP cdr(nil) is also equal to nil but under a LISP concept of equality; it may be
worthwhile trying to think LISP equality in behavioural abstraction terms.

10 Which may also have visible sorts in the arity.

This “neutrality” of cons with respect to the behavioural equivalence may be under-
stood by the fact that cons preserves the behavioural equivalence defined by cdr and
car only. This basic property of hidden constructors is called coherence [7, 4], which
in general means the preservation of the behavioural equivalence relation by the hid-
den constructors. In CafeOBJ the coherence property is user specified as an operation
attribute:

op cons : Elt List -> List {coherent}
The semantic meaning of a coherence declaration is that the corresponding spec-

ification admits only models for which the operation is coherent (i.e., it preserves the
behavioural equivalence). For methodological reason CafeOBJ admits potentially non-
coherent operations (in the absence of the coherence declaration), however in the final
version of the specification all hidden constructors should be declare coherent both for
semantical and operational reasons.11

2.3 Behavioural Coherence Methodologies

In the above list example the coherence of cons can be proved as a formal property of
the specification 12. This means that in any model of this specification the interpretation
of cons automatically preserves the behavioural equivalence, so the class of models
(implementations) of the specification with cons not specified as coherent coincides
with its subclass of models for the case when cons is specified as coherent. Such con-
structors, which occurs frequently and which are practically desirable are called the
conservative.

The opposite case is represented by the non-conservative constructors, which cor-
responds to the situation when the class of models for the case when the operation
is specified as coherent is a strict subclass of the class of models when the operation
is not specified as coherent. Proof-theoretically, this means the coherence property of
the operation cannot be formally proved as a consequence property of the [rest of the]
specification. Because of its semantical aspect, the methodology of non-conservative
constructors is more advanced and sophisticated than the conservative one. However it
might be very useful in dealing with non-terminating computations, in a way similar to
the the use of commutativity attribute for operations in classical algebraic specification
(see [4] for more details).

Proving Behavioural Coherence. We now concentrate to an example illustrating the
behavioural coherence methodology of conservative constructors. Consider the follow-
ing behavioural specification of sets:

11 A simple example is given by the coherence proof scores, when one needs to start with the
absence of the coherence declaration for the operation. The coherence declaration is added to
the specification only after its proof.

12 The CafeOBJ proof score for this is rather simple; we leave it as exercise for the reader.



Component-Based Algebraic Specification and Verification in CafeOBJ 1653

Elt

in
Bool

Set

empty

add

neg

U,&

E in empty = false
E in add(E’,S) = (E = E’) or (E in S)
E  in  (S1 U S2)  =  (E in S1) or (E in S2)
E  in  (S1 & S2)  =  (E in S1) and (E in S2)
E  in  neg(S)  =  not(E in S)

This specification has only one behavioural operation, namely the observation in . The
hidden constructors add, U , & , and neg can be proved coherent by the following
CafeOBJ proof score:

open .
ops s1 s2 s1’ s2’ : -> Set . -- arbitrary sets as temporary constants
ops e e’ : -> Elt . -- arbitrary elements as temporary constants
ceq S1 =*= S2 = true if (e in S1) == (e in S2) . -- definition of behavioural equivalence
beq s1 = s1’ . -- hypothesis
beq s2 = s2’ . -- hypothesis

red add(e, s1) =*= add(e, s1’) . -- beh coherence of add( ) for variable clash at Elt
red add(e’, s1) =*= add(e’, s1’) . -- beh coherence of add( ) for no variable clash at Elt
red (s1 U s2) =*= (s1’ U s2’) . -- beh coherence of U
red (s1 & s2) =*= (s1’ & s2’) . -- beh coherence of &
red neg(s1) =*= neg(s1’) . -- beh coherence of neg
close
where
Notice the simplicity of this proof score which uses the built-in default coinduc-

tion relation =*= which in practice is oftenly the behavioural equivalence. Once the
coherence of the hidden constructors is formally proved, their coherence declarations
are added to the specification, thus obtaining the final version of the specification under
the methodology of conservative hidden constructors.

2.4 Behavioural Verification

One of the great advantages of behavioural specification lies in the simplicity of the ver-
ification stage which sometimes contrasts sharply with the complexity of corresponding
data type verifications. Sets are one of the examples showing clearly the greater sim-
plicity of behavioural verifications. While the verification of set-theoretic properties in
the data approach gets into a very complex induction process, behavioural properties
of sets can be proved almost immediately. The following is the very simple CafeOBJ
proof score for one of De Morgan laws:

open .
op e : -> Elt .

ops s1 s2 s3 : -> Set .
-- definition of behavioural equivalence
ceq S1:Set =*= S2:Set = true if (e in S1) == (e in S2) .

red neg(s1 U s2) =*= (neg(s1) & neg(s2)) . -- proof of de Morgan law
close
Notice that CafeOBJ proof scores follow a certain pattern. Firstly, one “opens” an

working module for adding temporary entities; this is the command open. The com-
mand close “closes” the module, thus resetting it to its original content. After opening,
one introduces temporary constants which, due to the fact that they are arbitrary, have
the meaning of variables local to the proof score. The next step consists of setting up the
proof context, involving declaring the hypotheses, etc. The proofs are effectively done
by CafeOBJ reduction mechanism (the command red). The inputs of red are usually
logical atomic properties which are encoded as Boolean terms, so in general one ex-
pects a true answer from the CafeOBJ interpreter for each of such reductions. In case
of more complex proofs, a CafeOBJ proof score flatten the corresponding proof tree to
a sequence of hypotheses and reductions. In this flattening, properties proved by reduc-
tions might be lemmas which might be later used as hypotheses for other reductions.

Behavioural Rewriting. The execution of behavioural specifications is done by be-
havioural rewriting, which is a refinement of ordinary (term) rewriting that ensures
the correctness of rewriting when using behavioural equations as rewrite rules. The
basic condition of behavioural rewriting requires the existence of a path formed by be-
havioural or coherent operations on top of the redex. When inferring strict equalities,
it is required in addition that the top of such path is of visible sort. For example, when
proving the behavioural coherence of add,

red add(e, s1) =*= add(e, s1’) .
means a strict equality reduction. In this case the first behavioural equation of the cor-
responding proof score cannot be used as a first rewriting step since the condition of
behavioural rewriting is not fulfilled. This triggers the use of the conditional equation
instead as a first rewriting step, and only after this the use of behavioural equations of
the proof score fall under the required condition.

2.5 Behavioural Refinement

Object refinement in behavioural specification is a relaxed form of behavioural specifi-
cation morphism (see [7] for more details). As an example we show how behavioural
lists refine behavioural sets, which corresponds to the basic intuition of sets imple-
mented as lists. For simplicity of presentation we consider here only the case of basic
sets, without union, intersection, and negation13. The refinement of behavioural basic
sets to lists was represented above by extending the graphical notation previously intro-
duced with:

G3. Refinement of sorts and operations is written by / and sharing the same figure
(disk or arrow) in the diagram.

G4. Newly introduced sorts and operations are represented by dotted lines.

13 Our example can be easily extended to union and intersection, but not so easily to negation.

Set/List

Elt

?Elt

/cdr

/car

add/cons

in/...
Bool

/err

empty/nil

  
E  in  L  |---->  (E == car(L)) or-else
                       (car(L) =/= err) and-also (E in cdr(L)))

In this refinement, the hidden sort Set is refined to the hidden sort List (this means
that any state of the set object and be implemented by a state of the list object), add
is refined to cons. The list object has the observation car and the action cdr as new
behavioural operations and also adds the error handling. The set object observation in
is refined to a derived observation (using some operational versions of the Boolean
connectives). This refinement can be encoded in CafeOBJ by the following module
import:

mod* LIST’ { protecting(LIST)
op in : Elt List -> Bool {coherent} -- coherence provable from the rest of spec
vars E E’ : Elt
var L : List
eq E in L = (E == car(L)) or-else (car(L) =/= err and-also E in cdr(L)) . }

The following is the proof score for the fact that the mapping defined above is indeed
a refinement, i.e., the property of add holds for cons:14

open LIST’ .
ops e e1 e2 : -> Elt . -- arbitrary elements as temporary constants
op l : -> List . -- arbitrary list as temporary constant
eq e1 in l = true . -- the basic case when the element does belong to the list
eq e2 in l = false . -- the basic case when the element does not belong to the list

red e in nil == false . -- the nil case
red e1 in cons(e,l) == true .
red e2 in cons(e,l) == false .
red e in cons(e,l) == true . -- the element clash case
close

3 Concurrent Object Composition in CafeOBJ

In this section we present the object composition method of CafeOBJ based on the
behavioural specification paradigm. We present here a simplified method which does
not use behavioural coherence. We use UML to represent object composition:

14 This involves a small case analysis.

Object A

Object B

Object CObject D Object E

base level objects

In the above UML figure, B is composed of D and E, A of B and C, and non-
compound objects (i.e., objects with no components) are called base level objects. A
composition in UML is represented by line tipped by a diamond, and if necessary,
qualified by the numbers of components (1 for one and * for many).

Projection operations from the hidden sort of the states of the compound object
to the hidden sorts of the states of the component objects constitute the main techni-
cal concept underlying the CafeOBJ composition method; projection operations are
related to the lines of UML figures. Projection operations are subject to the following
mathematical conditions [8, 7]. Given a compound object O (with signature (V ∪H, S )
for which V is a set of visible sorts and H is a set of hidden sorts), a set of the labels of its
components Ob j, and the components {On}n∈Ob j of O (with signatures (Vn ∪Hn, S n)),
the projection operations p n : h→ hn (with h ∈H and hn ∈Hn) for each n ∈Ob j satisfy
the following conditions:

1. for each observation o of O, there exists n ∈ Ob j and a data (possibly derived)
operation f : vn1 ...vni → v (vni ∈Vn and v ∈V where i is a natural number), and for
each nk with k ∈ {1, . . . , i} a visible Onk -context cnk (i.e., made only of operations
of the signature of Onk) such that o(X) = f ((cn1 ◦ p n1)(X), ...,(cni ◦ p ni)(X)) for all
states X ,

2. for each action a of O, and for each n ∈ Ob j, there exists a sequence of actions an

such that p n◦a = an◦p n, and
3. for each constant const of O, and for each n ∈ Ob j, there exists a constant constn

such that p n(const) = constn (i.e., the initial state of a compound object should be
is related to the initial state of each component.)

In the compound objects we only define communication between the components; this
means that the only equations at the level of the specification of the compound objects
are the ones relating the actions and observations of the compound objects to those
of the components as described above. In the case of synchronized compositions, the
equations of the previous definition are conditional rather than unconditional. Their
conditions are subject to the following:

– each condition is a finite conjunction of equalities between terms of the form cn◦ p n

(where p n is a projection operator and cn is an On-context) and terms in the data
signature, and

– disjunction of all the conditions corresponding to a given left hand side (of equa-
tions regarded as a rewrite rule) is true.

3.1 Parallel Connection

The components of a composite object are connected (unsynchronized) in parallel if
there is no synchronization between them. In order to define the concept of synchro-
nization, we have to introduce the concept of action group. Two actions of a compound
object are in the same action group when they change the state of the same component
object via a projection operation. Synchronization appears when:

– there exists an overlapping between some action groups, or
– the projected state of the compound object (via a projection operation) depends on

the state of a different (from the object corresponding to the projection operation)
component.

The first case is sometimes called broadcasting and the second case is sometimes called
client-server computing. In the unsynchronized case, we have full concurrency between
all the components, which means that all the actions of the compound object can be ap-
plied concurrently, therefore the components can be implemented as distributed pro-
cesses or concurrent processes with multi-thread which are based on asynchronous
communications.

For unsynchronized parallel connection, we consider a bank account system exam-
ple. Firstly, we consider a very simple bank account system which consists of a fixed
numbers of individual accounts, lets actually consider the case of just two accounts.
The specification of an account can be obtained just by renaming the specification
COUNTER1 of a counter object with integers as follows

mod* ACCOUNT1 { protecting(COUNTER1 *{ hsort Counter -> Account1,
op init-counter -> init-account1 })}

mod* ACCOUNT2 { protecting(COUNTER1 *{ hsort Counter -> Account2,
op init-counter -> init-account2 })}

where COUNTER1 is represented in CafeOBJ graphical notation as follows:

Counter1

Int

add

amountamount(init-counter1) = 0
amount(add(I, C)) = I + amount(C)

init-counter

We then compose these two account objects as in the following double figure con-
taining both the UML and the CafeOBJ graphical15 representation of this composi-
tion, where deposit1 and withdraw1 are the actions for the first account, balance1 is
the observation for the first account, account1 is the projection operation for the first
account, and deposit2, withdraw2, balance2, and account2 are the corresponding
actions, observation, and projection operation for the second account:

AccountSys

Account1 Account2

deposit1

deposit2

withdraw1

withdraw2

add add

1

1

1

1

AccountSys

Account1 Account2

account1

deposit1
deposit2
withdraw1
withdraw2

balance1
balance2

account2

Nat

Int

init-account-sys

The equations for this parallel connection (composition) are as follows:
eq balance1(AS) = amount(account1(AS)) .
eq balance2(AS) = amount(account2(AS)) .
eq account1(init-account-sys) = init-account1 .
eq account1(deposit1(N, AS)) = add(N, account1(AS)) .
eq account1(deposit2(N, AS)) = account1(AS) .
eq account1(withdraw1(N, AS)) = add(-(N), account1(AS)) .
eq account1(withdraw2(N, AS)) = account1(AS) .
eq account2(init-account-sys) = init-account2 .
eq account2(deposit1(N, AS)) = account2(AS) .
eq account2(deposit2(N, AS)) = add(N, account2(AS)) .
eq account2(withdraw1(N, AS)) = account2(AS) .
eq account2(withdraw2(N, AS)) = add(-(N), account2(AS)) .

Notice that besides the first two equations relating the observations on the com-
pound object to those on the components, the other equations relate the actions of the
account system to the actions of the components. Remark that the actions corresponding
to one component do not change the state of the second component (via the projection
operation), hence this composition is unsynchronized. In fact these equations express-
ing the concurrency of composition need not be specified by the user, in their absence
they may be generated internally by the system, thus reducing the specification of the
composition to the essential information which should be provided by the user.

15 The CafeOBJ graphical representation corresponds to the module defining this object compo-
sition rather than to the “flattened” specification, hence the operations of the components are
not included in the figure.

3.2 Dynamic Connection

In this subsection, we extend the previous bank account system example to support an
arbitrary number of accounts. The accounts are created or deleted dynamically, so we
call such architecture pattern dynamic connection and we call the objects connected
dynamically as dynamic objects. A dynamic object has an object identifier type as the
arity of its initial state (which is quite a natural idea that in object-oriented programming
languages, language systems automatically providing a pointer for each object when
created). We therefore firstly extend the specification of the counter to a dynamic object

op init-counter : UId -> Counter
where UId is a sort for user identifiers. The actions add-account and del-account
maintain the user accounts. add-account creates accounts with some initial balance
while del-account deletes the accounts; both of them are parameterized by the user
identifiers UId. Each of deposit and withdraw is also parameterized by the user identi-
fiers. Most notably, the projection operation for Account is also parameterized by UId.
The initial state of AccountSys has no account, so it is mapped to the error state called
no-account. The structure of the new bank account system can be represented in UML
and CafeOBJ graphical notation as follows:

AccountSys

Account

add

1

*

add-account

del-account

deposit

withdraw

AccountSys Nat

deposit
withdraw

add-account
del-account

account

no-account

Account

UId

Finally, the equations relate the actions of AccountSys to those of Account via the pro-
jection operation only when they correspond to the specified user account. Here is the
essential part of the CafeOBJ code for the dynamic system of accounts specification:

eq account(U, init-account-sys) = no-account .
ceq account(U, add-account(U’, N, A)) = add(N, init-account(U)) if U == U’ .
ceq account(U, add-account(U’, N, A)) = account(U, A) if U =/= U’ .
ceq account(U, del-account(U’, A)) = no-account if U == U’ .
ceq account(U, del-account(U’, A)) = account(U, A) if U =/= U’ .
ceq account(U, deposit(U’, N, A)) = add(N, account(U, A)) if U == U’ .
ceq account(U, deposit(U’, N, A)) = account(U, A) if U =/= U’ .
ceq account(U, withdraw(U’, N, A)) = add(-(N), account(U, A)) if U == U’ .
ceq account(U, withdraw(U’, N, A)) = account(U, A) if U =/= U’ .

3.3 Synchronized Parallel Connection

In this subsection, we add a user database (UserDB) to the bank account system exam-
ple for having a more sophisticated user management mechanism. This enables query-

ing whether an user already has an account in the bank account system. The users data
base is obtained just by reusing (renaming) the set object of Section 12. The new ac-
count system compound object contains both synchronization patterns: broadcasting
and client-server computing. add-account is related to add of Account by the projec-
tion operation for Account and it is also related to add of UserDB by the projection
operation for UserDB. So, there is an overlapping of action groups (broadcasting).
Also, add-account is related to add of Account by the projection operation for Ac-
count using the information of UserDB (client-server computing). The same holds for
del-account. The following is the UML and CafeOBJ graphical representation of this:

AccountSys

UserDB Account

deposit

add-account

withdraw

del-account

add add

1

1

1

*

_U_

_&_

neg

AccountSys

UserDB Account

Nat UId

deposit
withdraw

add-account
del-account

accountuser-db

init-account-sys

and here is the CafeOBJ code representing the equations for the projection operation
for UserDB:

eq user-db(init-account-sys) = empty .
eq user-db(add-account(U, AS)) = add(U, user-db(AS)) .
eq user-db(del-account(U, AS)) = neg(add(U, empty)) & user-db(AS) .
eq user-db(deposit(U, N, AS)) = user-db(AS) .
eq user-db(withdraw(U, N, AS)) = user-db(AS) .

The following is the CafeOBJ code for the equations for the projection operation for
Account, we skip here the equations of deposit and withdraw which are the same as in
the previous example, and we also skip the equation for del-account which is similar
to that of add-account:

eq account(U, init-account-sys) = no-account .
ceq account(U, add-account(U’, N, AS)) = add(N, init-account(U))

if U == U’ and not(U in user-db(AS)) .
ceq account(U, add-account(U’, N, AS)) = account(U, AS)

if U =/= U’ or U in user-db(AS) .

For add-account, we check whether the user is already registered and if not map it to
add. If the user is already registered in UserDB, then skip.

It is interesting to mention that the same test
red balance(’u:UId, add-account(’u, 100, deposit(’u, 30, add-account(’u, 100,
init-account-sys))))

gets different results in the previous account system example and in the current syn-
chronized example due to the finer user management in the synchronized case.

3.4 Compositionality of Verifications

In object-oriented programming, reusability of the source code is important, but in
object-oriented specification, reusability of the proofs is also very important because of
the verification process. We call this compositionality of verifications of components. In
the CafeOBJ object composition method this is achieved by the following fundamental
Theorem (for its proof see [8]):

Theorem 1. Given the states s and s′ of a compound object then the behavioural equiv-
alence ∼ of the compound object is defined as follows:

s ∼ s′ iff p n(s) ∼n p n(s′) for all n ∈ Obj

where Ob j is a set of the labels for the components (you can consider these labels as
the names of the component modules), p n is the projection operation corresponding to
the component object n, and ∼n is the behavioural equivalence of the component.2

Therefore, in the case of a hierarchic object composition, the behavioural equivalence
for the whole system is just the conjunction of the behavioural equivalences of the base
level objects, which are generally rather simple.

For example, the behavioural equivalence for the bank account system is a con-
junction of the behavioural equivalence Account (indexed by the user identifiers) and
UserDB, and these two are checked automatically by the CafeOBJ system. This means
that behavioural proofs for the bank account system are almost automatic, without hav-
ing to go through the usual coinduction process. Therefore, the behavioural equivalence
R[ ] of AccountSys can be defined by the following CafeOBJ code:

mod BEQ-ACCOUNT-SYSTEM { protecting(ACCOUNT-SYSTEM)
op R[ ] : AccountSys UId AccountSys -> Bool
vars AS1 AS2 : AccountSys
var U : UId
eq AS1 R[U] AS2 = account(U, AS1) =*= account(U, AS2) and

user-db(AS1) =*= user-db(AS2) . }
Notice the use of the parameterized relation for handling the conjunction indexed by
the user identifiers.

Now, we will prove the true concurrency of withdrawals of two different users,
which can be considered as a safety property for this system of bank accounts and
which is formulated as the following commutativity behavioural property:

withdraw(u1,n1,withdraw(u2,n2,as)) ∼ withdraw(u2,n2,withdraw(u1,n1,as))

The following CafeOBJ code builds the proof tree containing all possible cases formed
by orthogonal combinations of atomic cases for the users with respect to their member-
ship to the user accounts data base. The basic proof term is TERM. The automatic
generation of the proof tree (RESULT )is done by a meta-level encoding in CafeOBJ
by using its rewrite engine for one-directional construction of the proof tree (this pro-
cess uses the rewriting logic feature of CafeOBJ, hence the use of transitions (trans)
rather than equations).
mod PROOF-TREE { protecting(BEQ-ACCOUNT-SYSTEM)

ops n1 n2 : -> Nat -- arbitrary amounts for withdrawal

ops u u1 u1’ u2 u2’ : -> UId -- arbitrary user identifiers
op as : -> AccountSys -- arbitrary state of the account system
eq u1 in user-db(as) = true . -- first user is in the data base
eq u2 in user-db(as) = true . -- second user is in the data base
eq u1’ in user-db(as) = false . -- first user is not in the data base
eq u2’ in user-db(as) = false . -- second user is not in the data base
vars U U1 U2 : UId
op TERM : UId UId UId -> Bool -- basic proof term
trans TERM(U, U1, U2) => withdraw(U1, n1, withdraw(U2, n2, as)) R[U]

withdraw(U2, n2, withdraw(U1, n1, as)) .
op TERM1 : UId UId -> Bool
trans TERM1(U, U1) => TERM(U, U1, u2) and TERM(U, U1, u2’) .
op TERM2 : UId -> Bool
trans TERM2(U) => TERM1(U, u1) and TERM1(U, u1’) .
op RESULT : -> Bool -- final proof term
trans RESULT => TERM2(u1) and TERM2(u1’) and TERM2(u) . }

The execution of the proof term RESULT gives true after the system performs 233
rewrites.

4 Conclusions and Future Work

In this paper we presented the CafeOBJ object-oriented methodology for component-
based specification and verification which is based on the CafeOBJ behavioural ab-
straction paradigm. We also presented the basic behavioural specification methodology
in CafeOBJ and gave a brief overview of the CafeOBJ language, system and specifi-
cation environment.

Future work in this area will further explore and refine the current CafeOBJ method-
ologies exposed here with the aim of creating an industrial tool around these method-
ologies containing an industrial-oriented tutorial, a GUI interface probably based on
the current CafeOBJ graphical notation, a graphical proof environment supporting dis-
tributed proofs over networks, etc.

Also, the power and sophistication of CafeOBJ gives the possibility to develop al-
ternative behavioural specification and verification methodologies, including component-
based ones. We plan to study such alternative methodologies and their relationship to
the current one.

